/**
 * @(#) ProfileProxy.as
 *
 * @date		2009. 02. 12.
 * @version		1.0
 * @author		Shin, Yunwoo
 * @description	 인사기록 조회 - Proxy
 * 				인사기록을 조회하는 Model Logic 
 */
package com.skt.common_hr.main.model
{
	import com.skt.common_hr.common.Variables;
	import com.skt.common_hr.common.puremvc.CommonProxy;
	import com.skt.common_hr.main.ApplicationFacade;
	import com.skt.common_hr.main.model.vo.AuthorityVO;
	import com.skt.common_hr.main.model.vo.LoginVO;
	import com.wcl.dataloader.*;
	
	import mx.collections.ArrayCollection;
	import mx.rpc.events.FaultEvent;
	import mx.rpc.events.ResultEvent;

	public class MyInfoProxy extends CommonProxy
	{
		public static const NAME:String = "ProfileProxy";
		
		public var profile:LoginVO;
		public var authority:ArrayCollection;
		public var pernr:String;

		/**
		 * 생성자
		 * 
		 * @param
		 * @return void
		 */
		public function MyInfoProxy(data:Object=null)
		{
			super(NAME, data);
		}
		
		
		/**
		 * data load를 수행
		 * 		- DataLoader 를 생성하고 Import Parameter 객체를 생성, RFC를 호출
		 * 
		 * @param pernr 사번
		 * @return void
		 */
		public function load(pernr:String):void
		{
			var param:Object = new Object(); 
			trace("MyInfoProxy.load");
			param["I_PERNR"] = this.pernr = pernr;
			if(Variables.company!='FNU') {
				param["I_YEA"] = Variables.I_YEA;	
			}
			 
			/* 
			if(Variables.company=='NOS') {
				param["I_YEA"] = Variables.I_YEA;	
			}
			 */
			//Alert.show('I_YEA : ' + Variables.I_YEA);
			
			loadData("Z_HR_0029", param, result, fault); 
		}
		
		
		
		
		
		
		
		///////////////////////////////////////////////////////////////////////////////////////////////////
		//
		//	Event Listeners
		//
		///////////////////////////////////////////////////////////////////////////////////////////////////
		/**
		 * 서버에서 정상적으로 리턴된 결과를 처리
		 * 
		 * @param
		 * @return void
		 */
		public function result(event:ResultEvent):void
		{
			if(retCode==0) {  
				profile = parseProfile(event.result.ITAB[0]);
				authority = parseAuthority(event.result.ATAB);
				facade.sendNotification(ApplicationFacade.LOAD_PROFILE_COMPLETE, this);
			}
			else {
				facade.sendNotification(ApplicationFacade.LOAD_PROFILE_FAILED, this);
			}
		}
		
		/**
		 * 서버에서 오류가 발생했을 처리
		 * 
		 * @param
		 * @return void
		 */
		private function fault(event:FaultEvent):void
		{
			facade.sendNotification(ApplicationFacade.LOAD_PROFILE_FAILED, this);
		}
		
		
		
		
		
		///////////////////////////////////////////////////////////////////////////////////////////////////
		//
		//	Data Parsing functions
		//
		///////////////////////////////////////////////////////////////////////////////////////////////////
		/**
		 * 서버에서 전달된 result 객체를 분석하여 Value Object를 생성
		 * 
		 * @param
		 * @return 인사기록 Value Object
		 */
		private function parseProfile(retObj:Object):LoginVO
		{
			if(retObj==null) return null;
			var vo:LoginVO = new LoginVO();
			vo.PERNR		= pernr;					// 사번
			vo.ENAME		= retObj["ENAME"];			// 성명
			vo.ORGEH		= retObj["ORGEH"];			// 부서코드
			vo.ORGTX		=retObj["ORGTX"];			// 부서명
			vo.CFPERNR		=retObj["CFPERNR"];			// 부서장 사번
			vo.CFNAME		=retObj["CFNAME"];			// 부서장 성명
			vo.PLANS		=retObj["PLANS"];			// 직책코드
			vo.PNAME		=retObj["PNAME"];			// 직책명
			vo.HWTOP		=retObj["HWTOP"];			// 사전지시근무
			
			vo.KOSTL		=retObj["KOSTL"];			// 코스트센터
			vo.CARDNO		=retObj["CARDNO"];			// 법인카드번호
			vo.SID			=retObj["SID"];				// 서버이름
			vo.TEL			=retObj["TEL"];				// 전화번호
			
			// ehrole
			return vo;
		}
		
		/**
		 * 서버에서 전달된 result 객체를 분석하여 Value Object를 생성
		 * 
		 * @param
		 * @return 인사기록 Value Object
		 */
		private function parseAuthority(src:ArrayCollection):ArrayCollection
		{
			var retArr:ArrayCollection = new ArrayCollection();
			for each(var rec:Object in src)
			{
				var vo:AuthorityVO = new AuthorityVO();
				vo.SEQNO		= rec["SEQNO"];
				vo.EMENU		= rec["EMENU"];			// 메뉴코드
				vo.EMENUTX		= rec["EMENUTX"];		// 메뉴명
				vo.EHROLE		= rec["EHROLE"];		// e-HR 권한구분
				vo.EMERL		= rec["EMERL"];			// 메뉴권한 구분(조회S/쓰기U/삭제D)
				vo.MANAGER		= rec["MANAGER"];		// 메뉴에 해당하는 관리자 사번
				vo.ENAME		= rec["ENAME"];			// 관리자 성명
				vo.ORGTX		= rec["ORGTX"];			// 관리자 부서
				vo.TEL_OFFICE	= rec["TEL_OFFICE"];	// 관리자 전화번호
				vo.FAX			= rec["FAX"];			// 관리자 fax
				vo.EMAIL_IN		= rec["EMAIL_IN"];
				Variables.myEhrole = vo.EHROLE;
				retArr.addItem(vo);
			}
			return retArr;
		}
	}
}